

# Formalization of Error-Correcting Codes: From Hamming to Modern Coding Theory

Reynald Affeldt<sup>1</sup>(✉) and Jacques Garrigue<sup>2</sup>

<sup>1</sup> National Institute of Advanced Industrial Science and Technology, Tsukuba, Japan  
`reynald.affeldt@aist.go.jp`

<sup>2</sup> Nagoya University, Nagoya, Japan  
`garrigue@math.nagoya-u.ac.jp`

**Abstract.** By adding redundancy to transmitted data, error-correcting codes (ECCs) make it possible to communicate reliably over noisy channels. Minimizing redundancy and (de)coding time has driven much research, culminating with Low-Density Parity-Check (LDPC) codes. At first sight, ECCs may be considered as a trustful piece of computer systems because classical results are well-understood. But ECCs are also performance-critical so that new hardware calls for new implementations whose testing is always an issue. Moreover, research about ECCs is still flourishing with papers of ever-growing complexity. In order to provide means for implementers to perform verification and for researchers to firmly assess recent advances, we have been developing a formalization of ECCs using the SSReflect extension of the Coq proof-assistant. We report on the formalization of linear ECCs, duly illustrated with a theory about the celebrated Hamming codes and the verification of the sum-product algorithm for decoding LDPC codes.

## 1 Introduction

Error-correcting codes (ECCs) add redundancy to transmitted data to ensure reliable communication over noisy channels. Low-Density Parity-Check (LDPC) codes are ECCs discovered in 1960 by Gallager; they were not thoroughly studied until they were shown in the nineties to deliver good performance in practice. Since then, LDPC codes have found their way into modern devices such as hard-disk storage, wifi communications, etc. and have motivated a new body of works known as modern coding theory.

Implementations of ECCs cannot be crystallized as a generic library that can be deemed correct because extensively tested. Because ECCs are performance-critical, new implementations are required to take advantage of the latest hardware, so that testing is always an issue. Also, research (in particular about LDPC codes) is so active that correctness guarantees for cutting-edge ECCs are scattered in scientific publications of ever-growing complexity.

A formalization of ECCs could help implementers and researchers. First, it would make possible verification of concrete implementations. Today, an implementer willing to perform formal verification should first provide a formal specification of what ECCs are supposed to achieve. In comparison, this is more

difficult than verification of cryptographic functions whose specification requires little infrastructure when they rely on number theory (e.g., [1]).

However, the formalization of ECCs is a difficult undertaking. They rely on a vast body of mathematics: probabilities, graphs, linear algebra, etc. Teaching material is rarely (if ever) structured as algebra textbooks: prose definitions that look incomplete without the accompanying examples, algorithms written in prose, hypotheses about the model that appear during the course of proofs, etc. Monographs and research papers do not provide details for the non-expert reader. It is therefore no wonder that researchers are seeking for means to firmly assess the correctness of their pencil-and-paper proofs: our work is part of such a project.

Still, there is previous work that we can take advantage of to formalize ECCs. The SSREFLECT/MATHCOMP library [8] provides big operators to formalize combinatorial results, a substantial formalization of linear algebra, and tools to reason about graphs. The formalization of the foundational theorems of information theory [2] provides us with basic definitions about channels and probabilities. Last, we are fortunate enough to have colleagues, expert in ECCs, who provided us with details about linear ECCs and LDPC codes [9, Chaps. 7 and 9].

To the best of our knowledge, our effort is the first attempt at a systematic formalization of ECCs inside a proof-assistant. In Sect. 3, we formalize basic results about linear ECCs. Already at this point, some effort was spent in augmenting textbook definitions with their implicit assumptions. In Sect. 4, we formalize Hamming codes. In particular, we provide a concrete encoder and decoder and express the error rate in terms of a closed formula. In Sect. 5, we formalize the key properties of the sum-product algorithm, the standard algorithm for efficient decoding of LDPC codes. Finally, in Sect. 6, we apply our formalization to the verification of a concrete implementation of the sum-product algorithm, making our work the first formal verification of a decoding algorithm for an advanced class of ECCs.

## 2 Premises on Information Theory and Probabilities

### 2.1 Channels and Codes in Information Theory

We first recall basic definitions from [2].

The most generic definition of a code is as a *channel code*: a pair of encoder/decoder functions with a finite type  $M$  for the message pieces to be encoded. Encoded message pieces (codewords) are represented by row vectors over a finite alphabet  $A$  (denoted by  $\text{'rV}[A]_{\text{'n}}$  in MATHCOMP). The decoder (that may fail) is fed with the outputs of a noisy channel that are also represented by row vectors (possibly over a different<sup>1</sup> alphabet  $B$ ):

```

Definition encT := {ffun M → 'rV[A]_n}.
Definition decT := {ffun 'rV[B]_n → option M}.
Record code := mkCode { enc : encT ; dec : decT }.

```

<sup>1</sup>  $A$  and  $B$  are different for example in the case of the *binary erasure channel* that replaces some bits with an *erasure*.

A (discrete) noisy channel is modeled as a stochastic matrix that we formalized as a function from the input alphabet  $\mathbf{A}$  to probability distributions over the output alphabet  $\mathbf{B}$ :

**Notation** " $\mathcal{CH}_1(\mathbf{A}, \mathbf{B})$ "  $:= (\mathbf{A} \rightarrow \text{dist } \mathbf{B})$ .

**dist** is the type of probability distributions. They are essentially functions from some finite type to non-negative reals whose outputs sum to 1 (the big operator  $\sum_{-}(\mathbf{x} \text{ in } \mathbf{P}) \mathbf{f } \mathbf{x}$  comes from MATHCOMP):

```
Record dist (A : finType) := mkDist {
  pmf :> A → R+ ; (* "→ R+" is a notation *)
  pmf1 :  $\sum_{-}(\mathbf{a} \text{ in } \mathbf{A}) \text{ pmf } \mathbf{a} = 1$  }.
```

Given a distribution  $\mathbf{P}$ , the probability of an event (represented by a finite set of elements: type  $\{\text{set } \mathbf{A}\}$  in MATHCOMP) is formalized as follows:

**Definition**  $\text{Pr } \mathbf{P} (\mathbf{E} : \{\text{set } \mathbf{A}\}) := \sum_{-}(\mathbf{a} \text{ in } \mathbf{E}) \mathbf{P } \mathbf{a}$ .

Communication of  $n$  characters is thought of as happening over the  $n^{\text{th}}$  extension of the channel, defined as a function from input vectors to distributions of output vectors ( $\{\text{dist } \mathbf{T}\}$  is a notation for the type of distributions; it hides a function that checks whether  $\mathbf{T}$  is a finite type):

**Notation** " $\mathcal{CH}_n(\mathbf{A}, \mathbf{B})$ "  $:= ('rV[\mathbf{A}]_n \rightarrow \{\text{dist } 'rV[\mathbf{B}]_n\})$ .

In this paper, we deal with *discrete memoryless channels* (DMCs). It means that the output probability of a character does not depend on preceding inputs. In this case, the definition of the  $n^{\text{th}}$  extension of a channel  $\mathbf{W}$  boils down to a probability mass function that associates to an input vector  $\mathbf{x}$  the following distribution of output vectors:

**Definition**  $\mathbf{f} (\mathbf{y} : 'rV_n) := \prod_{-}(\mathbf{i} < n) \mathbf{W} (\mathbf{x} /_{-} \mathbf{i}) (\mathbf{y} /_{-} \mathbf{i})$ .

where  $\mathbf{x} /_{-} \mathbf{i}$  represents the  $i$ th element of the vector  $\mathbf{x}$ . The notation  $\mathbf{W}^n(\mathbf{y} \mid \mathbf{x})$  ( $W^n(y|x)$  in pencil-and-paper proofs) is the probability for the DMC of  $\mathbf{W}$  that an input  $\mathbf{x}$  (of length  $n$ ) is output as  $\mathbf{y}$ .

Finally, the quality of a code  $\mathbf{c}$  for a given channel  $\mathbf{W}$  is measured by its *error rate* (notation:  $\bar{e}_{cha}(\mathbf{W}, \mathbf{c})$ ), that is defined by the average probability of errors:

```
Definition ErrRateCond (W :  $\mathcal{CH}_1(\mathbf{A}, \mathbf{B})$ ) c m :=
  Pr (W ^ n (| enc c m)) [set y | dec c y ≠ Some m].
Definition CodeErrRate (W :  $\mathcal{CH}_1(\mathbf{A}, \mathbf{B})$ ) c :=
  1 / INR #| M | *  $\sum_{-}(\mathbf{m} \text{ in } \mathbf{M}) \text{ ErrRateCond } \mathbf{W } \mathbf{c } \mathbf{m}$ .
```

$\mathbf{W}^n(\mathbf{y} \mid \text{enc } \mathbf{c } \mathbf{m})$  is the distribution of outputs corresponding to the codeword  $\text{enc } \mathbf{c } \mathbf{m}$  of a message  $\mathbf{m}$  sent over the DMC of  $\mathbf{W}$ .  $[\text{set } \mathbf{y} \mid \text{dec } \mathbf{c } \mathbf{y} \neq \text{Some } \mathbf{m}]$  is the set of outputs that do not decode to  $\mathbf{m}$ .  $\text{INR}$  injects naturals into reals.

## 2.2 Aposteriori Probability

Probabilities are used to specify the correctness of probabilistic decoders such as the sum-product algorithm (see Sect. 5).

We first formalize the notion of *aposteriori probability*: the probability that an input was sent knowing that some output was received. It is defined via the Bayes rule from the probability that an output was received knowing that some input was sent. For an input distribution  $P$  and a channel  $W$ , the aposteriori probability of an input  $x$  given the output  $y$  is:

$$P^W(x|y) := \frac{P(x)W^n(y|x)}{\sum_{x' \in A^n} P(x')W^n(y|x')}$$

We formalize aposteriori probabilities with the following probability mass function:

**Definition** `den` :=  $\sum (x \text{ in } \text{'rV\_n'}) P \ x * W \wedge \ n \ (y \mid x)$ .

**Definition** `f x` :=  $P \ x * W \wedge \ n \ (y \mid x) / \text{den}$ .

This probability is well-defined when the denominator is not zero. This is more than a technical hindrance: it expresses the natural condition that, since  $y$  was received, then necessarily a suitable  $x$  (i.e., such that  $P \ x \neq 0$  and  $W \wedge \ n \ (y \mid x) \neq 0$ ) was sent beforehand. The denominator being non-zero is thus equivalent to the `receivable` condition:

**Definition** `receivable y` :=  $[\exists \ x, (P \ x \neq 0) \wedge (W \wedge \ n \ (y \mid x) \neq 0)]$ .

In Coq, we denote aposteriori probabilities by  $P \wedge \wedge \ W, H \ (x \mid y)$  where  $H$  is a proof that  $y$  is `receivable`.

Finally, the probability that the  $n_0^{\text{th}}$  bit of the input is set to  $b$  (0 or 1) given the output  $y$  is defined by the *marginal aposteriori probability* ( $K$  is chosen so that it is indeed a probability):

$$P_{n_0}^W(b|y) := K \sum_{x \in \mathbb{F}_2^n, x_{n_0}=b} P^W(x|y)$$


In Coq, we will denote this probability by  $P \wedge \wedge \ n_0 \wedge \wedge \ W, H \ (b \mid y)$  where  $H$  is the proof that  $y$  is `receivable`. See [3] for complete formal definitions.

### 3 A Formal Setting for Linear ECCs

Linear ECCs are about bit-vectors, i.e., vectors over  $\mathbb{F}_2$  (we use the notation `'F_2` from MATHCOMP). Their properties are mostly discussed in terms of Hamming weight (the number of 1 bits) or of Hamming distance (the number of bits that are different). In Coq, we provide a function `wH` for the Hamming weight, from which we derive the Hamming distance: **Definition** `dH n (x y : 'rV_n)` := `wH (x - y)`.

#### 3.1 Linear ECCs as Sets of Codewords

The simplest definition of a linear ECC is as a set of codewords closed by addition ( $n$  is called the *length* of the code):



**Fig. 1.** The setting of error-correcting codes

```
Record lcode0 n := mkLcode0 {
  codewords :> {set 'rV['F_2]_n} ;
  lclosed : addr_closed codewords }.
```

In practice, a linear ECC is defined as the kernel of a *parity check matrix* (PCM), i.e., the matrix whose rows correspond to the checksum equations that codewords fulfill (\*m is multiplication and ^T is transposition of matrices):

```
Definition syndrome (H : 'M['F_2]_(m, n)) (y : 'rV_n) := H *m y^T.
Definition kernel H := [set c | syndrome H c = 0 ].
```

Since the kernel of the PCM is closed by addition, it defines a linear ECC:

```
Lemma kernel_add H : addr_closed (kernel H). Proof. ... Qed.
Definition lcode0_kernel H := mkLcode0 (kernel_add H).
```

When  $H$  is a  $m \times n$  matrix,  $k = n - m$  is called the *dimension* of the code.

A code is trivial when it is reduced to the singleton with the null vector:

```
Definition not_trivial := ∃ cw, (cw ∈ C) ∧ (cw ≠ 0).
```

When a linear ECC  $C$  is not trivial (proof  $C\_not\_trivial$  below), one can define the *minimum distance* between any two codewords, or, equivalently, the minimum weight of non-zero codewords, using SSREFLECT's `xchoose` and `arg_min` functions:

```
Definition non_0_codeword := xchoose C_not_trivial.
Definition min_wH_codeword :=
  arg_min non_0_codeword [pred cw in C | wH cw ≠ 0] wH.
Definition d_min := wH min_wH_codeword.
```

The minimum distance  $d_{\min}$  defines in particular the number of errors  $\lfloor \frac{d_{\min}-1}{2} \rfloor$  that one can correct using minimum distance decoding (see Sect. 3.3):

```
Definition mdd_err_cor := (d_min. - 1)/2.
```

### 3.2 Linear ECCs with Coding and Decoding Functions

In practice, a linear ECC is not only a set of codewords but also a pair of coding and decoding functions to be used in conjunction with a channel (see Fig. 1 [5, p. 16]). We combine the definition of a linear ECC as a set of codewords (Sect. 3.1) and as a pair of encoding and decoding functions (i.e., a channel code—Sect. 2.1) with the hypotheses that (1) the encoder is injective and (2) its image is a subset of the codewords:

```

Record lcode n k : Type := mkLcode {
  lcode0_of :> lcode0 n ;
  enc_dec :> code 'F_2' 'F_2' rV['F_2]_k n ;
  enc_inj : injective (enc enc_dec) ;
  enc_img : enc_img_in_code lcode0_of (enc enc_dec) }.

```

`enc_img_in_code` is the hypothesis that the image of the messages (here, `rV['F_2]_k`) by the encoder (`enc enc_dec`) is included in the set of codewords (here, `lcode0_of`). Note that  $k \leq n$  can be derived from the injectivity of the encoder.

As indicated by Fig. 1, the decoder is decomposed into (1) a function that repairs the received output and (2) a function that discards the redundancy bits:

```

Record lcode1 n k := mkLcode1 {
  lcode_of :> lcode n k ;
  repair : repairT n ; (*'rV['F_2]_n → option ('rV['F_2]_n) *)
  discard : discardT n k ; (*'rV['F_2]_n → 'rV['F_2]_k *)
  dec_is_repair_discard :
    dec lcode_of = [ffun y ⇒ omap discard (repair y)];
  enc_discard_is_id : cancel_on lcode_of (enc lcode_of) discard }.

```

`enc_discard_is_id` is a proof that `discard` followed by encoding `enc` is the identity over the domain `lcode_of`.

*Example.* The  $r$ -repetition code encodes one bit by replicating it  $r$  times. It has therefore two codewords:  $00 \dots 0$  and  $11 \dots 1$  ( $r$  times). The PCM can be defined as  $H = A \parallel 1$  where  $A$  is a column vector of  $r - 1$  1's, and the corresponding encoder is the matrix multiplication by  $G = 1 \parallel (-A)^T$ . More generally, let  $A$  be a  $(n - k) \times k$  matrix,  $H = A \parallel 1$  and  $G = 1 \parallel (-A)^T$ . Then  $H$  is the PCM of a  $(n, k)$ -code with the (injective) encoding function  $x \mapsto x \times G$ . Such a linear ECC is said to be in *systematic form* (details in [3]).

### 3.3 The Variety of Decoding Procedures

There are various strategies to decode the channel output. *Minimum distance decoding* chooses the closest codeword in terms of Hamming distance. When such a decoder decodes an output  $y$  to a message  $m$ , then there is no other message  $m'$  whose encoding is closer to  $y$ :

```

Definition minimum_distance_decoding :=
  ∀ y m, (dec c) y = Some m →
    ∀ m', dH ((enc c) m) y ≤ dH ((enc c) m') y.

```

We can now formalize the first interesting theorem about linear ECCs [11, p. 10] that shows that a minimum distance decoder can correct `mdd_err_cor` (see Sect. 3.1) errors:

```

Lemma encode_decode m y : (dec C) y ≠ None →
  dH ((enc C) m) y ≤ mdd_err_cor C_not_trivial →
  (dec C) y = Some m.

```

For example, a repetition code can decode  $(r - 1)/2$  errors (with  $r$  odd) since minimum distance decoding can be performed by majority vote (see [3] for formal proofs):

```

Definition majority_vote r (s : seq'F_2) : option'F_2 :=
  let cnt := num_occ 1\,s in
  if r/2 < cnt then Some 1
  else if (r/2 = cnt) ∧ ~~ odd r then None
  else Some 0.

```

*Maximum Likelihood (ML) Decoding* decodes to the message that is the most likely to have been encoded according to the definition of the channel. More precisely, for an encoder  $f$ , a ML decoder  $\phi$  is such that  $W^n(y|f(\phi(y))) = \max_{m \in M} W^n(y|f(m))$ :

```

Definition maximum_likelihood_decoding :=
  support (enc c) → ∀ y, receivable W P y →
  ∃ m, (dec c) y = Some m ∧
  W ^ n (y | (enc c) m) = \rmax_(m' in M) W ^
n (y | (enc c) m').

```

The assumption `receivable W P y` says that we consider outputs with non-zero probability (see Sect. 2.2). The assumption `support (enc c)` says that only codewords can be input. Textbooks do not make these assumptions explicit but they are essential to complete formal proofs.

ML decoding is desirable because it achieves the smallest error rate among all the possible decoders [3, Lemma ML\_smallest\_err\_rate]. Still, it is possible to achieve ML decoding via minimum distance decoding. This is for example the case with a binary symmetric channel (that inputs and outputs bits) with error probability  $p < \frac{1}{2}$ . Formally, for a code  $c$  with at least one codeword:

```

Lemma MD_implies_ML : p < 1/2 → minimum_distance_decoding c →
  (∀ y, (dec c) y ≠ None) → maximum_likelihood_decoding W c P.

```

*Maximum a posteriori probability (MAP) decoding* decodes to messages that maximize the a posteriori probability (see Sect. 2.2). MAP decoding is desirable because it achieves ML decoding [3, lemma MAP\_implies\_ML]. *Maximum posterior marginal (MPM) decoding* is similar to MAP decoding: it decodes to messages such that each bit maximizes the marginal a posteriori probability. The sum-product algorithm of Sect. 5 achieves MPM decoding.

## 4 Formalization of Hamming Codes and Their Properties

We formalize Hamming codes. In particular, we show that the well-known decoding procedure for Hamming code is actually a minimum distance decoding and that the error rate can be stated as a closed formula.

**Formal Definition.** Hamming codes are  $(n = 2^m - 1, k = 2^m - m - 1)$  linear ECCs, i.e., one adds  $m$  extra bits for error checking. The codewords are defined

by the PCM whose columns are the binary representations of the  $2^m - 1$  non-null words of length  $m$ . For a concrete illustration, here follows the PCM of the (7, 4)-Hamming code:

$$\text{ham}H_{7,4} = \begin{pmatrix} 0 & 0 & 0 & 1 & 1 & 1 & 1 \\ 0 & 1 & 1 & 0 & 0 & 1 & 1 \\ 1 & 0 & 1 & 0 & 1 & 0 & 1 \end{pmatrix}$$

Formally, for any  $m$ , we define the PCM using a function `nat2bin_cV` that builds column vectors with the binary representation of natural numbers (e.g., for the matrix  $H$  above, `nat2bin_cV 3 1` returns the first column vector, `nat2bin_cV 3 2` the second, etc.):

```
Definition hamH := \matrix_(i < m, j < n) (nat2bin_cV m j+1 i 0).
Definition hamC : lcode0 n := lcode0_kernel hamH.
```

**Minimum Distance.** The minimum distance of Hamming codes is 3, and therefore, by minimum distance decoding, Hamming codes can correct 1-bit errors (by the lemma `encode_decode` of Sect. 3.3). The fact that the minimum distance is 3 is proved by showing that there are no codewords of weights 1 and 2 (by analysis of  $H$ ) while there is a codeword of weight 3 ( $7 \times 2^{n-3} = (1110 \dots 0)_2$ ). Hamming codes are therefore not trivial and their minimum distance is 3:

```
Lemma hamming_not_trivial : not_trivial hamC.
Lemma minimum_distance_is_3 : d_min hamming_not_trivial = 3.
```

**Minimum Distance Decoding.** The procedure of decoding for Hamming codes is well-known. To decode the output  $y$ , compute its syndrome: if it is non-zero, then it is the binary representation of the index of the bit to flip back. The function `ham_detect` computes the index  $i$  of the bit to correct and prepare a vector to repair the error. The function `ham_repair` fixes the error by adding this vector:

```
Definition ham_detect y :=
  let i := bin2nat_cV (syndrome hamH y) in
  if i is 0 then 0 else nat2bin_rV n (2 ^ (n - i)).
Definition ham_repair : decT _ _ m := [ffun y =>
  let ret := y + ham_detect y in
  if syndrome hamH ret = 0 then Some ret else None].
```

Let `ham_score` be a linear ECC using the `ham_repair` function. We can show that it implements minimum distance decoding:

```
Lemma hamming_MD : minimum_distance_decoding ham_score.
```

It is therefore an ML decoding (by the Lemma `MD_implies_ML` from Sect. 3.3).

**The Encoding and Discard Functions.** We now complete the formalization of Hamming codes by providing the encoding and discard functions (as in Fig. 1). Modulo permutation of the columns, the PCM of Hamming codes can be transformed into systematic form  $\text{sys}H = \text{sys}A \parallel 1$  (as explained in the example about repetition codes in Sect. 3.2). This provides us with a generating matrix  $\text{sys}G = 1 \parallel (-\text{sys}A)^T$ . For illustration in the case of the (7, 4)-Hamming code:



$$\text{sys}H_{7,4} = \left( \begin{array}{ccc|ccc} 0 & 1 & 1 & 1 & 1 & 0 & 0 \\ 1 & 0 & 1 & 1 & 0 & 1 & 0 \\ 1 & 1 & 0 & 1 & 0 & 0 & 1 \end{array} \right) \quad \text{sys}G_{7,4} = \left( \begin{array}{ccc|ccc} 1 & 0 & 0 & 0 & 0 & 1 & 1 \\ 0 & 1 & 0 & 0 & 1 & 0 & 1 \\ 0 & 0 & 1 & 0 & 1 & 1 & 0 \\ 0 & 0 & 0 & 1 & 1 & 1 & 1 \end{array} \right)$$

Let `sysH_perm` be the column permutation that turns `H` into `sysH`. The parity check and generating matrices in systematic form are formalized as follows:

**Definition** `sysH` : 'M['F\_2]\_(m, n) := `col_perm sysH_perm hamH`.

**Definition** `sysG` :=

`castmx (erefl, subnK (m_len m')) (row_mx 1%M (-sysA)^T)`.

(`castmx` is a cast that deals with dependent types.) The discard function in systematic form is obvious:

**Definition** `sysDiscard` : 'M['F\_2]\_(n - m, n) :=

`castmx (erefl, subnK (m_len m')) (row_mx 1%M 0)`.

Using the column permutation `sysH_perm` the other way around, we can produce the discard function and the generating matrix corresponding to the original `hamH`:

**Definition** `ham_discard` := `col_perm sysH_perm^-1 sysDiscard`.

**Definition** `hamG` := `col_perm sysH_perm^-1 sysG`.

Coupled with the `ham_repair` function above, `hamG` and `ham_discard` provides us with a complete definition of Hamming encoders and decoders:

**Definition** `ham_channel_code` := `mkCode`

`[ffun t => t *m hamG] [ffun x =>`

`omap ham_discard (ham_repair _ x)]`.

**Error Rate.** Finally, we show, in the case of a binary symmetric channel `W`, that the error rate (see Sect. 2.1) of Hamming codes can be expressed as a closed formula:

**Lemma** `hamming_error_rate` : `p < 1/2 →`

`ēcha(W, ham_channel_code) =`

`1 - ((1 - p) ^ n) - INR n * p * ((1 - p) ^ (n - 1))`.

The existence of codes with arbitrary small error rates is the main result of Shannon's theorems. But Shannon's proofs are not constructive. Our formalization of Hamming codes with a closed formula for their error rate provides us with a concrete candidate.

## 5 Formalization of the Properties of Sum-Product Decoding

The sum-product algorithm provides efficient decoding for LDPC codes. It computes for each bit its marginal aposteriori probability by propagating probabilities in a graph corresponding to the PCM. We explain those graphs in Sect. 5.1, the summary operator used to specify the sum-product algorithm in Sect. 5.2, and the main properties of the sum-product algorithm in Sect. 5.3.

### 5.1 Parity Check Matrix as Tanner Graphs

The vertices of a Tanner graph correspond to the rows and columns of a parity-check matrix  $H$  with an edge between  $m$  and  $n$  when  $H_{m,n} = 1$ . Rows are called *function nodes* and columns are called *variable nodes*. By construction, a Tanner graph is bipartite.

Sets of successor nodes and *subgraphs* of Tanner graphs appear as indices of big operators in the definitions and proofs of the sum-product algorithm.

Let  $g$  be a graph (formalized by a binary relation) and  $m$  and  $n$  be two connected vertices. The subgraph rooted at the edge  $m$ - $n$  is the set of vertices reachable from  $m$  without passing through  $n$ :

```
Variables (V : finType) (g : rel V).
Definition except n := [rel x y | g x y ∧ (x ≠ n) ∧ (y ≠ n)].
Definition subgraph m n :=
[set v | g n m ∧ connect (except n) m v].
```

For Tanner graphs, we distinguish successors and subgraphs of variable nodes and of function nodes. We denote the successors of the function (resp. variable) node  $m_0$  (resp.  $n_0$ ) by  $\text{'V } m_0$  (resp.  $\text{'F } n_0$ ). We denote the function nodes of the subgraph rooted at edge  $m_0$ - $n_0$  by  $\text{'F}(m_0, n_0)$ . Similarly, we denote the variable nodes of the subgraph rooted at edge  $m_0$ - $n_0$  (to which we add  $n_0$ ) by  $\text{'V}(m_0, n_0)$ . Figure 2 provides an explanatory illustration, see [3] for complete definitions.




Fig. 2. Successors and subtrees in an acyclic Tanner graph

It will be important to distinguish acyclic Tanner graphs:

```
Definition acyclic g := ∀ l, 2 < size l → ~ path.ucycle g l.
```

Technically, we will need to establish partition properties when proving the properties of the sum-product decoding algorithm (see Sect. 5.3 for a concrete example).

### 5.2 The Summary Operator

Pencil-and-paper proofs in modern coding theory [12] make use of a special summation called the *summary operator* [10]. It is denoted by  $\sum_{\sim_s}$  and indicates

the variables *not* being summed over. This operator saves the practitioner “from a flood of notation” [12, p. 49], for example by writing steps such as:

$$\prod_{m_0 \in F(n_0)} \sum_{\sim \{n_0\}} \cdots = \sum_{\sim \{n_0\}} \prod_{m_0 \in F(n_0)} \cdots, \quad (1)$$

the reader being trusted to understand that both operators sum over different sets.

We formalize the summary operator as a sum over vectors  $\mathbf{x}$  such that  $\mathbf{x} \_ i$  is fixed using a default vector  $\mathbf{d}$  when  $i \notin s$  and write  $\sum_{-}(\mathbf{x} \# s, \mathbf{d})$  instead of  $\sum_{\sim s}$ :

**Definition** `summary (s : {set 'I_n}) (d x : 'rV[A]_n) :=`

`[ $\forall i, (i \in \sim : s) \Rightarrow (\mathbf{x} \_ i = \mathbf{d} \_ i)$ ].`

**Notation** `" $\sum_{-} ( \mathbf{x} \# s, \mathbf{d} ) e$ " := ( $\sum_{-} ( \mathbf{x} \mid \text{summary } s \mathbf{d} \mathbf{x} ) e$ )`

Indeed,  $\sum_{\sim s}$  can be understood as a sum over vectors  $[x_0; x_1; \dots; x_{n-1}]$  such that  $x_i$  is fixed when  $i \in s$ . We found it difficult to recover the terseness of the pencil-and-paper summary operator in a proof-assistant. First, the precise set of varying  $x_j$  is implicit; it can be inferred by looking at the  $x_j$  appearing below the summation sign but this is difficult to achieve unless one reflects most syntax. Second, it suggests working with vectors  $x$  of varying sizes, which can be an issue when the size of vectors appears in dependent types (tuples or row vectors in MATHCOMP). Last, it is not clear about the values of  $x_i$  when  $i \in s$ .

In contrast, our formalization makes clear, for example, that in Eq. (1) the first summary operator sums over  $V(m_0, n_0) \setminus \{n_0\}$  while the second one sums over  $[1, \dots, n] \setminus \{n_0\}$  (see Sect. 5.3 for the formalization). More importantly, we can benefit from MATHCOMP lemmas about big operators to prove the properties of the sum-product decoding thanks to our encoding (see Sect. 5.3).

Alternatively, our summary operator  $\sum_{-}(\mathbf{x} \# s, \mathbf{d}) \mathbf{e} \mathbf{x}$  can also be thought as  $\sum_{x_1 \in \mathbb{F}_2} \cdots \sum_{x_{|s|} \in \mathbb{F}_2} e[d[s_1 := x_1] \cdots [s_{|s|} := x_{|s|}]]$  where  $d[i := b]$  represents the vector  $d$  where index  $i$  is updated with  $b$ . Put formally (enum  $s$  below is the list  $[s_1; s_2; \dots; s_{|s|}]$ ):

**Definition** `summary_fold (s : {set 'I_n}) d e :=`

`foldr (fun n0 F t  $\Rightarrow \sum_{-} (b \text{ in } 'F_2) F (t '[n0 := b])$ ) e (enum s) d.`

This is equivalent ( $\sum_{-}(\mathbf{x} \# s, \mathbf{d}) \mathbf{e} \mathbf{x} = \text{summary\_fold } s \mathbf{d} \mathbf{e}$ ) but we found it easier to use `summary_fold` to prove our implementation of the sum-product algorithm in Sect. 6.

### 5.3 Properties of the Sum-Product Decoding

**Correctness of the Estimation.** Let us consider a channel  $W$  and a channel output  $y$ . With sum-product decoding, we are concerned with evaluating  $P_{n_0}^W(b|y)$  where  $b$  is the value of the  $n_0^{\text{th}}$  bit of the input codeword (see Sect. 2.2). In the following, we show that it is proportional to the following quantity:

$$P_{n_0}^W(b|y) \propto W(y_{n_0}|b) \prod_{m_0 \in F(n_0)} \alpha_{m_0, n_0}(b).$$

$\alpha_{m_0, n_0}(b)$  (formal definition below) is the contribution to the marginal aposteriori probability of the  $n_0^{\text{th}}$  bit coming from a subtree of the Tanner graph (we assume that the Tanner graph is acyclic).

We now provide the formal statement. Let  $W$  be a channel. Let  $H$  be a  $m \times n$  PCM such that the corresponding Tanner graph is acyclic (hypothesis `acyclic_graph (tanner_rel H)`, where `tanner_rel` turns a PCM into the corresponding Tanner graph). Let  $y$  be the channel output to decode. We assume that it is receivable (hypothesis  $H_y$  below, see Sect. 2.2). Finally, let  $d$  be the vector used in the summary operator. Then the aposteriori probability  $P_{n_0}^W(b|y)$  can be evaluated by a closed formula:

```
Lemma estimation_correctness (d : 'rV_n) n0 :
  let b := d /_ n0 in let P := 'U C_not_empty in
  P ' _ n0 ' ^ W, Hy (b | y) =
  Kmpp Hy * Kpp W H y * W b (y /_ n0) *  $\prod$  (m0 in 'F n0)  $\alpha$  m0 n0 d.
```

`Kmpp` and `Kpp` are normalization constants (see [3]).  $P$  is a uniform distribution. The distribution `'U C_not_empty` of codewords has the following probability mass function:  $cw \mapsto 1/|C|$  if  $cw \in C$  and 0 otherwise.  $\alpha$  is the marginal aposteriori probability of the  $n_0^{\text{th}}$  bit of the input codeword in the modified Tanner graph that includes only function nodes from the subgraph rooted at edge  $m_0 - n_0$  and in which the received bit  $y /_ n0$  has been erased. The formal definition relies on the summary operator:

```
Definition  $\alpha$  m0 n0 d :=  $\sum$  (x # 'V(m0, n0) : \ n0 , d)
  W ^ _ (y # 'V(m0, n0) : \ n0 | x # 'V(m0, n0) : \ n0) *
   $\prod$  (m1 in 'F(m0, n0)) INR ( $\delta$  ('V m1) x).
```

$\delta s x$  is an indicator function that performs checksum checks:

```
Definition  $\delta$  n (s : {set 'I_n}) (x : 'rV['F_2]_n) :=
  (\big[+%R/Zp0]_ (n0 in s) x /_ n0) = Zp0.
```

Let us comment about two technical aspects of the proof of `estimation_correctness`. The first one is the need to instrument Tanner graphs with partition lemmas to be able to decompose big sums/prods. See the next paragraph on lemma `recursive_computation` for a concrete example. The second one is the main motivation for using the summary operator. We need to make big sums commute with big prods in equalities like:

$$\prod (m_0 \text{ in } 'F \ n_0) \sum (x \# 'V(m_0, n_0) : \ n_0 , d) \dots = \sum (x \# \text{setT} : \ n_0 , d) \prod (m_0 \text{ in } 'F \ n_0) \dots$$

Such steps amount to apply the MATHCOMP lemma `big_distr_big_dep` together with technical reindexing. This is one of our contributions to provide lemmas for such steps.

**Recursive Computation of  $\alpha$ 's.** The property above provides a way to evaluate  $P_{n_0}^W(b|y)$  but not an efficient algorithm because the computation of  $\alpha_{m_0, n_0}(b)$  is about the whole subgraph rooted at the edge  $m_0 - n_0$ . The second property that we formalize introduces  $\beta$  probabilities such that  $\alpha$ 's (resp.  $\beta$ 's) can be computed from neighboring  $\beta$ 's (resp.  $\alpha$ 's). This is illustrated by Fig. 3 whose meaning will be made clearer in Sect. 6. We define  $\beta$  using  $\alpha$  as follows:

**Definition**  $\beta$   $n_0$   $m_0$  ( $d : 'rV\_n$ ) :=  
 $W (d \ /\_ n_0) (y \ /\_ n_0) * \prod_{-}(m_1 \text{ in } 'F\ n_0 : \setminus m_0) \alpha\ m_1\ n_0\ d.$

We prove that  $\alpha$ 's can be computed using  $\beta$ 's by the following formula (we assume the same setting as for the Lemma `estimation_correctness`):

**Lemma** `recursive_computation`  $m_0$   $n_0$   $d : n_0 \in 'V\ m_0 \rightarrow$   
 $\alpha\ m_0\ n_0\ d = \sum_{-}(x \ \# \ 'V\ m_0 : \setminus n_0, d)$   
 $INR (\delta ('V\ m_0)\ x) * \prod_{-}(n_1 \text{ in } 'V\ m_0 : \setminus n_0) \beta\ n_1\ m_0\ x.$


This proof is technically more involved than the lemma `estimation_correctness` but relies on similar ideas: partitions of Tanner graphs to split big sums/prods and commutations of big sums and big prods using the summary operator. Let us perform the first proof step for illustration. It consists in turning the inner product of  $\alpha$  messages  $\prod_{-}(m_1 \text{ in } 'F(m_0, n_0) : \setminus m_0) INR (\delta ('V\ m_1)\ x)$  into:

$$\prod_{-}(n_1 \text{ in } 'V\ m_0 : \setminus n_0) \prod_{-}(m_1 \text{ in } 'F\ n_1 : \setminus m_0) \prod_{-}(m_2 \text{ in } 'F(m_1, n_1)) INR (\delta ('V\ m_2)\ x)$$

This is a consequence of the fact that  $'F(m_0, n_0) : \setminus m_0$  can be partitioned (when  $H$  is acyclic) into smaller  $'F(m_1, n_1)$  where  $n_1$  is a successor of  $m_0$  and  $m_1$  is a successor of  $n_1$ , i.e., according to the following partition:

**Definition** `Fgraph_part_Fgraph`  $m_0$   $n_0 : \{\text{set } \{\text{set } 'I\_m\}\} :=$   
 $(\text{fun } n_1 \Rightarrow \bigcup_{-}(m_1 \text{ in } 'F\ n_1 : \setminus m_0) 'F(m_1, n_1)) \ @ : (( 'V\ m_0) : \setminus n_0).$

Once `Fgraph_part_Fgraph`  $m_0$   $n_0$  has been shown to cover  $'F(m_0, n_0) : \setminus n_0$  with pairwise disjoint sets, this step essentially amounts to use the lemmas `big_trivIset` and `big_imset` from `MATHCOMP`. See [3, `tanner_partition.v`] for related lemmas.



**Fig. 3.** Illustrations for `sumprod_up` and `sumprod_down`. Left: `sumprod_up` computes the up links from the leaves to the root. Right: `sumprod_down` computes the down link of edge  $m_0-n_2$  using the  $\beta$ 's of edges  $m_0-n_i$  ( $i \neq 2$ ).

## 6 Implementation and Verification of Sum-Product Decoding

An implementation of sum-product decoding takes as input a Tanner graph and an output  $y$ , and computes for all variable nodes, each representing a bit of the

decoded codeword, its marginal aposteriori probability. One chooses to decode the  $n_0^{\text{th}}$  bit either as 0 if  $P_{n_0}^W(0|y) \geq P_{n_0}^W(1|y)$  or as 1 otherwise, so as to perform MPM decoding.

The algorithm we implement is known in the literature as the forward/backward algorithm and has many applications [10]. It uses the tree view of an acyclic Tanner graph to structure recursive computations. In a first phase it computes  $\alpha$ 's and  $\beta$ 's (see Sect. 5.3) from the leaves towards the root of the tree, and then computes  $\alpha$ 's and  $\beta$ 's in the opposite direction (starting from the root that time). Figure 3 illustrates this.

Concretely, we provide Coq functions to build the tree, compute  $\alpha$ 's and  $\beta$ 's, and extract the estimations, and prove formally that the results indeed agree with the definitions from Sect. 5.3.

**Definition of the Tree.** Function nodes and variable nodes share the same data structure, and are just distinguished by their kind.

```

Definition R2 := (R * R)%type.
Inductive kind : Set := kf | kv.
Fixpoint negk k := match k with kf => kv | kv => kf end.
Inductive tag : kind -> Set := Func : tag kf | Var : R2 -> tag kv.
Inductive tn_tree (k : kind) (U D : Type) : Type :=
  Node { node_id : id; node_tag : tag k;
        children : seq (tn_tree (negk k) U D);
        up : U; down : D }.

```

This tree is statically bipartite, thanks to the switching of the kind for the children. Additionally, in each variable node, `node_tag` is expected to contain the channel probabilities for this bit to be 0 or 1, i.e., the pair  $(W(y_{n_0}|0), W(y_{n_0}|1))$ . The `up` and `down` fields are to be filled with the values of  $\alpha$  and  $\beta$  (according to the kind), going to the parent node for `up`, and coming from it for `down`. Here again we will use pairs of the 0 and 1 cases. Note that the values of  $\alpha$ 's and  $\beta$ 's need not be normalized.

**Computation of  $\alpha$  and  $\beta$ .** The function  $\alpha_\beta$  takes as input the tag of the source node, and the  $\alpha$ 's and  $\beta$ 's from neighboring nodes, excluding the destination, and computes either  $\alpha$  or  $\beta$ , according to the tag. Thanks to this function, the remainder of the algorithm keeps a perfect symmetry between variable and function nodes.

```

Definition alpha_op (out inp : R2) :=
  let (o,o') := out in let (i,i') := inp in
  (o*i + o'*i', o*i' + o'*i).
Definition beta_op (out inp : R2) :=
  let (o,o') := out in let (i,i') := inp in (o*i, o'*i').
Definition alpha_beta k (t : tag k) : seq R2 -> R2 :=
  match t with
  | Func => foldr alpha_op (1,0)
  | Var v => foldl beta_op v
  end.

```

The definition for  $\beta$  is clear enough: assuming that  $\mathbf{v}$  contains the channel probabilities for the corresponding bit, it suffices to compute the product of these probabilities with the incoming  $\alpha$ 's. For  $\alpha$ , starting from the `recursive_computation` lemma, we remark that assuming a bit to be 0 leaves the parity unchanged, while assuming it to be 1 switches the parities. This way, the sum-of-products can be computed as an iterated product, using  `$\alpha_{\text{op}}$` . This optimization is described in [10, Sect. 5-E]. We will of course need to prove that these definitions compute the same  $\alpha$ 's and  $\beta$ 's as in Sect. 5.3.

**Propagation of  $\alpha$  and  $\beta$ .** `sumprod_up` and `sumprod_down` compute respectively the contents of the `up` and `down` fields.

```
Fixpoint sumprod_up {k} (n : tn_tree k unit unit)
  : tn_tree k R2 unit :=
  let children' := map sumprod_up (children n) in
  let up' :=  $\alpha_{\beta}$  (node_tag n) (map up children') in
  Node (node_id n) (node_tag n) children' up' tt.
Fixpoint seqs_but1 (a b : seq R2) :=
  if b is h::t then (a++t)::seqs_but1 (rcons a h) t else [].
Fixpoint sumprod_down {k} (n : tn_tree k R2 unit)
  (from_above : option R2) : tn_tree k R2 R2 :=
  let (arg0, down') :=
    if from_above is Some p then ([::p], p) else ([::], (1, 1)) in
  let args := seqs_but1 arg0 (map up (children n)) in
  let funs := map
    (fun n' l  $\Rightarrow$  sumprod_down n' (Some ( $\alpha_{\beta}$  (node_tag n) l)))
    (children n) in
  let children' := apply_seq funs args in
  Node (node_id n) (node_tag n) children' (up n) down'.
Definition sumprod {k} n := sumprod_down (@sumprod_up k n) None.
```

The `from_above` argument is `None` for the root of the tree, or the  $\beta$  coming from the parent node otherwise. `apply_seq` applies a list of functions to a list of arguments. This is a workaround to allow defining `sumprod_down` as a `Fixpoint`.

**Building the Tree.** A parity-check matrix  $H$  and the probability distribution  $\mathbf{rW}$  for each bit (computed from the output  $\mathbf{y}$  and the channel  $\mathbf{W}$ ) is turned into a `tn_tree`, using the function `build_tree`, and fed to the above `sumprod` algorithm:

```
Variables (W :  $\mathcal{CH}_1$ ('F_2, B)) (y : 'rV[B]_n).
Let rW n0 := (W 0 (y /_ n0), W 1 (y /_ n0)).
Let computed_tree := sumprod (build_tree H rW (k := kv) ord0).
```

**Extraction of Estimations.** We finally recover normalized estimations from the tree:

```
Definition normalize (p : R2) :=
  let (p0, p1) := p in (p0 / (p0 + p1), p1 / (p0 + p1)).
Fixpoint estimation {k} (n : tn_tree k R2 R2) :=
  let l := flatten (map estimation (children n)) in
  if node_tag n is Var _ then
```

```

(node_id n, normalize ( $\beta_{\text{op}}$  (up n) (down n))) :: 1
else 1 (* node_tag n is Func *).

```

**Correctness.** The correctness of the algorithm above consists in showing that the estimations computed are the intended a posteriori probabilities:

```

Let estimations := estimation_computed_tree.
Definition esti_spec n0 (x : 'rV_n) :=
  ('U C_not_empty) ' _ n0 ^^ W, Hy (x / _ n0 | y).
Definition estimation_spec := uniq (unzip1 estimations) ^
  ∀ n0, (inr n0, p01 (esti_spec n0) n0) ∈ estimations.

```

where  $p01\ f\ n0$  applies  $f$ , to a vector whose  $n0^{\text{th}}$  bit is set to 0 and 1.

Theorem `estimation_ok` in [3, `ldpc_algo_proof.v`] provides a proof of this fact. As key steps, it uses the lemmas `recursive_computation` and `estimation_correctness` from Sect. 5.3.

**Concrete Codes.** All proofs of probabilistic sum-product decoding assume the Tanner graph to be acyclic [10]. In practice codes based on acyclic graphs are rare and not very efficient [7]. We tested our implementation with one of them [3, `sumprod_test.ml`].

In the general case where the Tanner graph contains cycles, one would use an alternative algorithm that computes the  $\alpha$ 's and  $\beta$ 's repeatedly, propagating them in the graph until the corrected word satisfies the parity checks, failing if the result is not reached within a fixed number of iterations [10, Sect. 5]. This works well in practice but there is no proof of correctness, even informal. In place of this iterative approach, one could also build a tree approximating the graph, by unfolding it to a finite depth, and apply our functional algorithm.

## 7 Related Work

Coding theory has been considered as an application of the interface between the Isabelle proof-assistant and the Sumit computer algebra system [4]. In order to take advantage of the computer algebra system, proofs are restricted to a certain code length. Though the mathematical background about polynomials has been formally verified, results about coding theory are only asserted. In comparison, we formally verify much more (generic) lemmas. Yet, for example when proving that certain bitstrings are codewords, we found ourselves performing formal proofs close to symbolic computation. With this respect, we may be able in a near future to take advantage of extensions of the MATHCOMP library that provide computation [6].

## 8 Conclusion

In this paper, we have proved the main properties of Hamming codes and sum-product decoding. It is interesting to contrast the two approaches, respectively known as classical and modern coding theory.



For Hamming codes, we could provide an implementation of minimal-distance decoding, and prove that it indeed realizes maximum likelihood decoding, i.e., the best possible form of decoding.

For sum-product decoding, which provides the basis for LDPC codes, one can only prove that the sum-product algorithm allows to implement MPM decoding. However, this is two steps away from maximum likelihood: the proof is only valid for acyclic Tanner graphs, while interesting codes contain cycles, and MPM is an approximation of MAP decoding, with only the latter providing maximum likelihood. Yet, this “extrapolation” methodology does work: sum-product decoding of LDPC codes is empirically close to maximum likelihood, and performs very well in practice.

**Acknowledgments.** T. Asai, T. Saikawa, K. Sakaguchi, and Y. Takahashi contributed to the formalization. The formalization of modern coding theory is a collaboration with M. Hagiwara, K. Kasai, S. Kuzuoka, and R. Obi. The authors are grateful to the anonymous reviewers for their comments. This work is partially supported by a JSPS Grant-in-Aid for Scientific Research (Project Number: 25289118).

## References

1. Affeldt, R., Nowak, D., Yamada, K.: Certifying assembly with formal security proofs: the case of BBS. *Sci. Comput. Program.* **77**(10–11), 1058–1074 (2012)
2. Affeldt, R., Hagiwara, M., Sénizergues, J.: Formalization of Shannon’s theorems. *J. Autom. Reason.* **53**(1), 63–103 (2014)
3. Affeldt, R., Garrigue, J.: Formalization of error-correcting codes: from Hamming to modern coding theory. Coq scripts. <https://staff.aist.go.jp/reynald.affeldt/ecc>
4. Ballarin, C., Paulson, L.C.: A pragmatic approach to extending provers by computer algebra—with applications to coding theory. *Fundamenta Informaticae* **34**(1–2), 1–20 (1999)
5. Betten, A., Braun, M., Fripertinger, H., Kerber, A., Kohnert, A., Wassermann, A.: *Error-Correcting Linear Codes-Classification by Isometry and Applications*. Springer, Heidelberg (2006)
6. Dénès, M., Mörtberg, A., Siles, V.: A refinement-based approach to computational algebra in Coq. In: Beringer, L., Felty, A. (eds.) *ITP 2012. LNCS*, vol. 7406, pp. 83–98. Springer, Heidelberg (2012)
7. Etzion, T., Trachtenberg, A., Vardy, A.: Which codes have cycle-free Tanner graphs? *IEEE Trans. Inf. Theory* **45**(6), 2173–2181 (1999)
8. Gonthier, G., Mahboubi, A., Tassi, E.: A small scale reflection extension for the Coq system. Technical report RR-6455, INRIA (2008). Version 14, March 2014
9. Hagiwara, M.: *Coding theory: mathematics for digital communication*. Nippon Hyoron Sha (2012) (in Japanese)
10. Kschischang, F.R., Frey, B.J., Loeliger, H.-A.: Factor graphs and the sum-product algorithm. *IEEE Trans. Inf. Theory* **47**(2), 498–519 (2001)
11. MacWilliams, F.J., Sloane, N.J.A.: *The Theory of Error-Correcting Codes*. North-Holland, Amsterdam (1977). 7th impression (1992)
12. Richardson, T., Urbanke, R.: *Modern Coding Theory*. Cambridge University Press, Cambridge (2008)

Interactive Theorem Proving

6th International Conference, ITP 2015, Nanjing, China,

August 24-27, 2015, Proceedings

Urban, C.; Zhang, X. (Eds.)

2015, XI, 469 p. 63 illus., Softcover

ISBN: 978-3-319-22101-4